

B

Appendix

如何利用 VBA 來存取 Access 資料庫的記錄

B-1 ADO 資料存取架構

VBA 透過 ADO 的架構來存取 Access 資料庫。ADO 的架構如圖（B-1-1）所示，它是 Microsoft 設計來讓前端應用程式與資料庫之間的標準存取介面。任何的應用程式透過這個標準的介面，都可以去存取符合這個介面的資料庫。




圖 B-1-1 ADO 資料庫存取架構

VBA 透過 Connection 物件連接至 Access 資料庫。與資料庫連接成功之後，接下來就用 Command 物件或 Recordset 物件來對 Access 資料庫下達 SQL 指令來做記錄的存取。

Command 物件不回傳回結果集，所以可以利用它來新增、修改或刪除記錄。

Recordset 物件會傳回結果集，所以可以利用它來讀取記錄。

B-2 連接至資料庫

要連接至資料庫，我們必須建立 Connection 物件。建立 Connection 物件的程式碼如下：

```
'定義連接物件
Dim cnn As ADODB.Connection
'建立連接物件
Set cnn = New ADODB.Connection
'開啟物件
cnn.Open "Provider=Microsoft.ACE.OLEDB.12.0; " & _
"Persist Security Info=False; " & _
"User ID=Admin; " & _
"Data Source=" & CurrentProject.Path & _
"\客戶; "
'關閉連接物件
cnn.Close
'釋放連接物件
Set cnn = Nothing
```

但是，我們不用這麼麻煩。由於我們都是在 Access 2010 裡來撰寫資料庫存取的程式碼，而 Access 2010 自己也是透過 ADO 去存取自己所建立的資料庫檔案。所以，我們只要取得目前 Access 2010 已經建立的連接物件即可，不用大費周章的自己去建立 Connection 連接物件。例如以下的程式碼就可以取得目前 Access 2010 專案的連接物件。

```
CurrentProject.Connection
```

B-3 從 Access 資料庫讀取記錄出來

要自 Access 2010 讀取記錄出來，我們必須先建立讀取記錄（SELECT）的 SQL 指令，然後建立 RecordSet 物件，然後再利用 RecordSet 物件的 Open 方法來透過 Connection 物件來連接至 Access 2010 資料庫以執行 SQL 指令所要求的動作。

由於 SELECT 命令會取得資料，所以我們取得的記錄會放在我們所定義的 RecordSet 物件裡。

例如，讀取客戶資料表的程式碼可以寫成這樣：

```
'定義文字字串的變數
Dim strSQL As String
'輸入取得出貨單當日最大的出貨單號碼的查詢SQL語法
strSQL = "SELECT * " & _
        "FROM 客戶 " & _
        "WHERE 客戶編號 = '020' "
'定義資料集物件RecordSet的物件變數
Dim rst As ADODB.Recordset
'建立資料集物件RecordSet
Set rst = New ADODB.Recordset
'開啟資料集物件，以連接至KINGS資料庫
rst.Open strSQL, CurrentProject.Connection

If Not rst.EOF Then
    查詢結果 = rst.Fields(0) + rst.Fields(1)
End If

'關閉資料集物件Recordset
rst.Close
'釋放資料集物件Recordset所佔用的資源
Set rst = Nothing
```

B-4 新增記錄

要新增記錄至資料表，我們必須先建立新增記錄（INSERT）的 SQL 指令，然後建立 Command 物件，再利用 Command 物件的 Execute 方法來透過 Connection 物件來連接至 Access 2010 資料庫以執行新增 SQL 指令的動作。

例如，新增一筆客戶記錄到客戶資料表的程式碼可以寫成這樣：

```
'定義文字字串的變數
Dim strSQL As String
'定義命令物件Command的物件變數
Dim cmd As ADODB.Command
'建立命令物件Command
Set cmd = New ADODB.Command

'命令物件取得資料庫連接
cmd.ActiveConnection = CurrentProject.Connection
'定義新增記錄的SQL指令給字串變數
```

```

strSQL = "INSERT INTO 客戶 (客戶編號, 公司名稱, " & _
        "統一編號, 連絡人, 職稱, 電話, 傳真, " & _
        "郵遞區號, 公司地址) VALUES ('020', '我的公司', " & _
        "'12345678', '陳惟彬', '工友', '02-12345678', " & _
        "'02-87654321', '123', '台北市中正一路1號') "

'指定命令物件的SQL敘述
cmd.CommandText = strSQL
'執行Command物件
cmd.Execute

```

B-5 修改記錄

要修改資料表內的記錄，我們必須先建立修改記錄（UPDATE）的 SQL 指令，然後再建立 Command 物件，再利用 Command 物件的 Execute 方法來透過 Connection 物件來連接至 Access 2010 資料庫以執行修改記錄的 SQL 指令。

例如，修改客戶資料表的記錄，程式碼可以寫成這樣：

```

'定義文字字串的變數
Dim strSQL As String
'定義命令物件Command的物件變數
Dim cmd As ADODB.Command
'建立命令物件Command
Set cmd = New ADODB.Command

'命令物件取得資料庫連接
cmd.ActiveConnection = CurrentProject.Connection

'定義新增記錄的SQL指令給字串變數
strSQL = "UPDATE 客戶 SET 公司名稱 = '星羽有限公司', " & _
        "統一編號 = '27297949', 連絡人 = '陳星羽', " & _
        "職稱 = '總經理', 電話 = '(02)27857416', " & _
        "傳真 = '(02)27857729', 郵遞區號 = '98765', " & _
        "公司地址 = '台北市松山區大雅路12號' " & _
        "WHERE 客戶編號 = '020' "

'指定命令物件的SQL敘述
cmd.CommandText = strSQL
'執行Command物件
cmd.Execute

```

B-6 刪除記錄

要刪除資料表內的記錄，我們必須先建立刪除記錄（DELETE）的 SQL 指令，然後再建立 Command 物件，再利用 Command 物件的 Execute 方法來透過 Connection 物件來連接至 Access 2010 資料庫以執行刪除記錄的 SQL 指令。

例如，刪除客戶資料表的記錄，程式碼可以寫成這樣：

```
'定義文字字串的變數
Dim strSQL As String
'定義命令物件Command的物件變數
Dim cmd As ADODB.Command
'建立命令物件Command
Set cmd = New ADODB.Command

'命令物件取得資料庫連接
cmd.ActiveConnection = CurrentProject.Connection

'定義新增記錄的SQL指令給字串變數
strSQL = "DELETE FROM 客戶 " & _
        "WHERE 客戶編號 = '020' "

'指定命令物件的SQL敘述
cmd.CommandText = strSQL
'執行Command物件
cmd.Execute
```

